الفصل الأول

جمل التكرار Looping

سنشرح في هذا الفصل استخدام جمل التكرار (Loops)
ونعنى بها تلك العبارات التى تساعد في تحديد الشروط واتخاذ القرارات في حالة وجود أكثر من بديل، وتلك التى تستخدم في التكرارات والدوارات بقصد تقليل تعليمات الإجراء.
بالانتهاء من هذا الفصل ستكتسب المعارف وتتدرب على المهارات التى تجعلك قادرا على:

- ***جملة التكرار (الدوارة) For...Next***
- ***جملة التكرار (الدوارة) Do...Loop***
- ***جملة التكرار (الدوارة)* Do… While**
- ***جملة التكرار (الدوارة)* Do…Until**
- ***جملة التكرار (الدوارة) For…Each***
- **مثال تطبيقي علي العبارات الشرطية والدورات**

الدوارات Loops

لعل قوة الكمبيوتر تكمن أكثر ما تكمن في قدرته على تكرار المهام دون كلل أو خطأ، والتكرار هو أساس معظم العمليات الهندسية و العلمية و كذلك المعلوماتية والمحاسبية كأن تطبق وظيفة ما على كل سجلات قاعدة بيانات ، مثل طباعة مرتبات العاملين أو حساب الخصومات على كافة الموظفين و هكذا. وتعرف الكلمات الخاصة التي توضع في بداية و نهاية العبارات المطلوب تكرارها بالدوارات.

يستخدم Visual Basic نوعين من الدوارات، الأولى بعداد و تسمى Counter loop وهي التي تنفذ عدد محدد من المرات، الثانية لا يحدد فيها العدد و إنما تعتمد على أحد الشروط وتسمى الدوارات المشروطةConditional loops . فيما يلي شرح لهذين النوعين بالتفصيل.

الدوارة For...Next

الدوارات ذات العداد، هي الدوارات المسماة For أو For Next و ذلك لأن الجزء المراد تكرار تنفيذه يكتب بين كلمتي For و Next .

الصورة العامة للعبارة For كالتالي:

```
For counter = startval To endval [ step stepval ]
    Statement1
    Statement2
    ……
Next [counter]
```

كما ترى يتم تحديد متغير يسمى العداد ثم نحدد القيمة المبدئية للعداد و القيمة النهائية ، ويمكننا أن نحدد الخطوة step التي يتم زيادة العداد بها في كل دورة ، إن لم نحدد هذه الخطوة فالقيمة الافتراضية أن العداد يزيد بمقدار ١ في كل مرة ، فإذا تعدى العداد القيمة endvalue فإن تنفيذ الدوارة يتوقف وينتقل التنفيذ إلى أول جملة بعد Next . إذا كانت القيمة النهائية أصغر من القيمة المبدئية فإن الدوارة لن تنفذ على الإطلاق ، إلا إذا تم تحديد قيمة سالبة للخطوة ، مثلا step –1 .

عند استخدام العداد يجب أن تتنبه إلى حدود قيم المتغير، فمثلا المتغير من النمط Byte أقصى قيمة يسعها هي ٢٥٥، وعليه لا يمكنك استخدامه كعداد يعد حتى ٤٠٠ مثلا ، و عليك حينئذ استخدام عداد من نمط آخر.

لا تقم بتغيير قيمة العداد داخل الدوارة أبداً، قد يؤدي ذلك إلى دوارة لا نهائية infinite loop وذلك بأن يظل البرنامج يكرر هذه الدوارة بلا نهاية مسببا عطل للنظام.

أحيانا نحتاج إلى الخروج من الدوارة لسبب معين كان تنتهي وظيفة التكرار. يمكننا ذلك باستخدام Exit For ، التي تنقل التنفيذ إلى التعليمية التي تلي Next مباشرة.

مثال ١ :

المثال الآتي يستخدم الدوارة For…Next في أبسط صورها لطباعة الأرقام من ١ إلى ٥

```
Dim i As Integer
```

```
For i = 1 To 5
   MessageBox.Show(i)
Next i
```

وعن هذا المثال نوضح مايلى:

- أعلنا عن المتغير (I) للحصول على أرقام متغيرة من 1 إلى 5.
- تعمل الدوارة For .. Next على زيادة قيمة المتغير بمقدار (١) فى كل مرة دون الحاجة إلى كتابة الأمر التالى فى كل دورة.

```
i = i + 1
```

وذلك لأن الشرط المحدد فى أمر For وهو i=1 To 5 يشتمل على بداية عداد الدوارة ونهايته، حيث أن i=1 معناها بداية العداد المستخدم داخل الدوارة ، To 5 نهاية عداد الدوارة. ويجب الالتزام بهذه الصيغة دائما مع أمر For...Next أى كتابة بداية العداد بعد علامة = ونهايته بعد كلمة To.

- يتم الخروج تلقائياً من الأمر For .. Next بمجرد الوصول إلى العدد (5) فى المتغير (I) وهذا يعنى أن عدد مرات التكرار التى تمت لمجموعة الأوامر المطلوب تكرارها هو (5).

استخدام الوظيفة Step()

يتم استخدام الوظيفة Step() مع الأمر For .. Next لمعرفة مقدار الزيادة التى تتم على المتغير فى كل دورة. ففى الحالة السابقة لم نذكر الوظيفة Step() فى الأمر:

```
For i = 1 To 5
```

لأن الزيادة التلقائية فى كل دورة مقدارها ١مالم تذكر خلاف ذلك بالوظيفة Step() ، أى أن الأمر السابق يساوى تماما هذا الأمر:

```
For i = 1 To 5 Step 1
```

فمثلا لطباعة الأرقام الزوجية فقط فى الأرقام ١ الى ١٠، استخدم الكود التالى:

```
For i = 2 To 10 Step 2
Debug.WriteLine(i)
next i
```

إذا قمت بتنفيذ هذا الكود، تحصل على النتيجة الموضحة بشكل ١-١ التالى.

شكل ١–١ طباعة الأرقام الزوجية فقط باستخدام الوظيفة Step.

مثال ٢ :

المثال الآتي مثال هام، حيث يبين كيفية استخدام الدوارات ، وكيفية الخروج الطارئ منها، هذا خلافا لكونه ذو فائدة عملية. المثال يبحث في مصفوفة حرفية باسم NameArray عن اسم معين هو stsearch ، وبمجرد العثور على هذا الاسم، يتم الخروج من الدوارة وينتقل التنفيذ للسطر التالي لها.وبالطبع معظم عمليات البحث تتم هكذا، حيث لا حاجة أن نكمل البحث بعد العثور على ما نريده.

```
For cntr = 1 To 30
        found=Instr(1, NameArray(Index), stsearch,1)
        If found>0 then
                txtResult.Text=NameArray(cntr)
                Exit For
        End If
Next cntr
```

لاحظ في المثال السابق أن الدالة Instr تستخدم لاختبار وجود stsearch في أي عنصر من عناصر المصفوفة، وإن وجد فإن الدالة ترجع الحرف الذي يبدأ عنده النص المطلوب (١ أو أكثر) و ترجع الدالة صفرا إن لم تجد هذا النص. بعد ذلك يتم اختبار الناتج، فإن كان أكبر من صفر فإن العنصر الذي تم العثور عليه يتم عرضه في مربع نص يسمىtxtResult ، ثم يتم الخروج من الدوارة من خلال العبارة Exit For ، ذلك لأننا لا نحتاج لتكملة البحث بعد أن وجدنا ما نريد. المثال السابق أيضا يوضح العلاقة الوثيقة بين المصفوفات والدوارات، حيث يمكن القيام بمهام عالية الكفاءة بالدمج بينهما.

نذكر أن Visual Basic وخلاف العديد من اللغات الأخرى تتيح استخدام متغير كسري Single أو Double كمتغير للدوارة كما تتيح استخدام خطوة Stepكسرية (أقل من ١) ولكن يجب ألا نلجأ إلى هذه الطريقة إلا عند الحاجة الشديدة لأن الأعداد الكسرية أصعب في التعامل واكتشاف الأخطاء من الأعداد الصحيحة.

الدوارة Do... Loop

الفرق الرئيسي بين الدوارات التكرارية والدوارات المشروطة أن الأخيرة تعتمد في تنفيذها على شرط (condition) ، بينما الأولى تنفذ عدد معين من المرات.

الشرط الذي يعتمد عليه التنفيذ هو أي تعبير يمكن أن ينتج القيمة True أو False . هذا التعبير من الممكن أن يكون دالة مثل Eof (والتي تحدد هل وصلت إلى نهاية ملف أم لا) أو قيمة أحد خواص كائن مثل الخاصية Value الخاصة بزر الخيارات Option Button، أو تعبير به مقارنة مثل inAge > 25 ، أو متغير من النوع Boolean . هناك نوعين من الدوارات المشروطة هما Do While و Do Until وسنشرحهما بالتفصيل في الفقرة التالية.

دوارة Do... While

يتم تنفيذ الدوارة Do While طالما ظل الشرط متحققا، وعندما لا يتحقق، يتم نقل التنفيذ إلى العبارة التالية للدوارة.

هناك طريقتين أو شكلين لكتابة هذه الدوارة، كالتالي:

الشكل الأول

```
Do While condition
        statement
        ……
Loop
```

الشكل الثاني

```
Do
        statement
        ……
Loop While condition
```

الفرق بين الشكلين هو "متى يتم اختبار تحقق الشرط"، في الشكل الأول الاختبار يتم قبل إجراء التعليمات في الدوارة، وعليه فلو كان الشرط غير متحقق لن يتم تنفيذ العبارات في الدوارة ولا مرة. بينما في الشكل الثاني يتم الاختبار بعد إجراء التعليمات في الدوارة وعليه فإن هذه التعليمات تنفذ مرة واحدة على الأقل حتى ولو كان الشرط غير متحقق. واحد

فقط من الشكلين يمكن استخدامه، أي لا يمكن وضع **While** في أول وآخر الحلقة. واستخدام أي منهما يعتمد على التطبيق والوظيفة التي تريدها، وكلا النوعين قد يكون مفيد لك.

في الإصدارات القديمة من **Visual Basic** كانت الكلمة **Wend** تستخدم بدلا من **Loop** و لا يزال **Visual Basic** يدعمها حتى الآن إلا أنه من المستحسن استخدام الطريقة الجديدة في كتابة الدوارة.

مثال٣:

التعليمات التالية تعطى نفس نتيجة المثال الذي شرحناه لطباعة الأرقام من ١ الى ٥ :

```
Dim i As integer
i = 1
Do While i < = 5
        Debug.WriteLine(i)
        i = i + 1
Loop
```

***الدوارة* Do...Until**

تعد الدوارة **Do Until** عكس الدوارة السابقة **Do While** ، حيث أن تنفيذ الدوارة هنا يستمر ما لم *يتحقق* الشرط فإن تحقق يتم الخروج من الدوارة إلى التعليمة التالية. واستخدامها مماثل تماما للدوارة السابقة **Do While** ، و هناك أيضا شكلين لكتابتها كالتالي:

الشكل الأول:

```
Do Until condition
        statement
        ……
Loop
```

الشكل الثاني:

```
Do
        statement
        ……
Loop Until condition
```

و لهما نفس المعنى السابق، حيث أن الشكل الثاني ينفذ مرة على الأقل قبل الخروج من الحلقة. أكثر استخدام للدوارة **Do Until** هو معالجة الملفات ذات السجلات المتعاقبة، أو قواعد البيانات.

مثال ٤ :

التعليمات التالية تعطى نفس نتيجة المثال السابق:

```
Dim i As integer
i = 1
Do Until i > 5
        Debug.WriteLine(i)
        i = i + 1
Loop
```

لاحظ الفرق بين المثالين السابقين تجده يتركز فى الصيغة:

```
Do While i <= 5
```

ومعناه أنه سيتم التكرار طالما أن المتغير **(i)** أصغر من أو يساوى القيمة **5** بينما الصيغة التالية:

```
Do UNTIL i > 5
```

تعني أنه سيتم التكرار حتى تصبح قيمة المتغير **(i)** أكبر من خمسة.

لاحظ أن العلاقة **(=>)** هى عكس العلاقة **(<)**.

على أية حال الدوارات المشروطة أكثر مرونة من الدوارات ذات العداد. ويمكن تحويل ذات العداد إلى دوارة مشروطة بأن نستخدم متغير ولكن في هذه الحالة يجب أن نقوم بزيادة قيمته بأنفسنا، الأمر الذي يتم تلقائيا في حالة الدوارات ذات العداد.

من المشاكل المصاحبة للدوارات مشكلة الدوارات اللانهائية و التي تنتج إذا أخطأت في تحديد الشروط، فتظل الدوارة تعمل إلى لا نهاية مسببة عطل تام للنظام. لكي تنجو من ذلك الفخ إن كنت تعمل في بيئة **Visual Basic** اضغط **Ctrl+Break** لإيقاف عمل البرنامج. أما إن حدث ذلك في برنامج تم ترجمته إلى **EXE** وتقوم بتشغيله مستقلا، فإن الحل للخروج من هذه الحالة (دون الاضطرار لإعادة تشغيل الجهاز) أن تقوم بضغط **Ctrl+Alt+Delete** ليظهر لك مربع مدير المهام **Task Manager** الذى يمكنك من خلاله إغلاق البرنامج. والذي يمكنك من خلاله إغلاق البرنامج المتعطل.

الدوارة For...Each

يمكنك استخدام الدوارة For... Each لتنفيذ عدد من العبارات على مجموعة من الكائنات المرتبطة، لذا فهى غالباً ما تستخدم مع المصفوفات Arrays ومجموعات البيانات DataSet وأدوات التحكم. للتعرف على طريقة عمل هذه الدوارة، دعنا نرى الكود التالى:

```
Dim MyNumber As Integer
Dim MyArray() As Integer = {5,7,9,3,1,6,8}
For Each MyNumber In MyArray
    If Number > 5 Then Debug.WriteLine(MyNumber)
Next MyNumber
```

حيث يتم اختبار جميع عناصر المصفوفة ومن ثمَّ طباعة الأرقام الأكبر من ٥ (انظر شكل ١-٢).

شكل ١-٢ استخدام الدوارة For Each.

ومن أهم استخدامات الدوارة For Each، تنفيذ مجموعة من العمليات على عدد من أدوات التحكم الموجودة بالنموذج. فعلى سبيل المثال، لإخفاء جميع أدوات التحكم الموجودة بالنموذج الحالى، يمكنك استخدام الكود التالى:

```
Dim CtrlVal As Control
For Each CtrlVal In Controls
CtrlVal.Visible = False
Next CtrlVal
```

وقد قمنا فى هذا الكود بدايةً بتعريف المتغير CtrlVal الذى سيحتوى على أدوات التحكم ثم استخدام هذا المتغير داخل الدوارة، التى يتم فيها تخصيص القيمة False للخاصية Visible المصاحبة لكل أداة من أدوات التحكم الموجودة بالنموذج.

مثال تطبيقي علي العبارات الشرطية والدورات

بعد الأمثلة البسيطة التى قدمناها لكل نوع من العبارات السابقة، نوضح الآن مثالا أكبر وأكثر فائدة يحتوي على العديد من العبارات التي ذكرناها. المثال الذي نقدمه هو إنشاء عارض ملفات نصوص Text والتي لها الاختصار txt والتي ينشئها محرر النصوص في Windows وتجده على القرص المدمج المرفق بالكتاب باسم TextView .

إنشاء واجهة عارض الملفات

١. قم بإنشاء مشروع نوافذى جديد ثم اجعل اسمه TextView .

٢. غير اسم النموذج الرئيسي إلى frmView من خلال الخاصية Name وغير عنوانه إلى "عارض الملفات" من خلال الخاصية Text وقم أيضاً بتغيير قيمة الخاصية Right to Left إلى Yes.

٣. أضف مربع نص Text Box إلى النموذج وحدد خصائصه كالتالي:

```
Name = txtView
MultiLine = True
ScrollBars = Both
```

٤. أضف أداة مربع فتح الملفات OpenFileDialog من مربع الأدوات إلى النموذج، تظهر الأداة أسفل النموذج كالمعتاد.

٥. قم بإعادة تسمية الأداة من خلال الخاصية Name إلى opnText .

٦. أضف زر أمر وقم بتغيير عنوانه إلى "فتح" و اسمه إلى btnOpen .

يجب أن يظهر النموذج الآن كما فى شكل ١-٣ التالى.



شكل ١–٣ مظهر النموذج عند إضافة عناصر الواجهة الأساسية إليه.

إضافة الكود للبرنامج

كود البرنامج بسيط للغاية، انقر الزر "فتح" نقراً مزدوجاً ثم قم بإدخال الكود التالى داخل إجراء النقر الخاص بالزر:

```
Dim F As String 'اسم الملف
Dim N As Integer     ' رقم الملف
Dim S As String      ' مخزن لسطر
Dim Buffer As String ' مخزن للملف بالكامل
OpnText.Filter = "Text Files|*.txt"
OpnText.ShowDialog()
F = OpnText.FileName
Me.Text = F      ' تغيير عنوان النافذة
N = FreeFile()
FileOpen(N, F, OpenMode.Input)
Do Until EOF(N)
        S = LineInput(N)
        Buffer = Buffer & vbCrLf & S
Loop
txtView.Text = Buffer
FileClose(N)
```

يستخدم البرنامج أربعة متغيرات كما ترى تعليقا عليها بجانبها. وكافة المهام يتم تنفيذها عند ضغط زر btnOpen ولذا تجدها في الإجراء Click لهذا الزر. يبدأ الإجراء بإعداد الخاصية Filter لمربع الحوار Open بحيث تظهر الملفات النصية txt فقط. بعد ذلك يتم فتح المربع و اختيار اسم الملف ووضعه في المتغير F ومن ثمَّ وضع اسم الملف على شريط عنوان النموذج. بعد ذلك يتم حجز رقم متاح باستخدام الدالة FreeFile()، يلي ذلك فتح الملف من خلال الدالة FileOpen() ثم ملء المتغير Buffer بسطور الملف النصي وذلك باستخدام الدوارة:

```
Do Until EOF(N)
        S = LineInput(N)
        Buffer = Buffer & vbCrLf & S
Loop
```

كما ترى استخدمنا دوارة مشروطة Do Until ووضعنا الشرط في البداية لأن الملف قد لا يحتوي على أية سطور، الشرط هنا هو EOF(N) أو العثور على علامة انتهاء الملف. بداخل الدوارة سطرين الأول يقوم بقراءة سطر من الملف في المتغير S باستخدام الدالة LineInput() والثاني يقوم بإضافة هذا السطر (مع علامة بداية السطر CR وسطر جديد LF) ونستخدم لهما الثابت VbCrLf إلى المتغير Buffer . هنا قد يتبادر سؤال للذهن: لماذا لا نكتب مباشرة في مربع النصوص بدلا من استخدام Buffer ؟ ، للإجابة على هذا السؤال قم بإعادة تنفيذ البرنامج بدون Buffer وستجد فرقا هائلا في سرعة تنفيذ البرنامج، يرجع ذلك إلى أن الوصول إلى المتغير البسيط أسرع بمراحل من الوصول إلى خاصية في أداة تحكم. أخيراً يتم كتابة المتغير Buffer في مربع النصوص وإغلاق الملف من خلال الدالة FileClose() (انظر شكل ١-٤).



شكل ١-٤ نافذة برنامج عارض الملفات أثناء عمله وبه أحد الملفات النصية.

هناك قصور في هذا البرنامج و هو أنك إذا لم تختر ملف من مربع الحوار فسيحدث خطأ. للتغلب على هذه المشكلة نستخدم العبارة الشرطية **If** كالتالي:

```
If F = " " Then
    MessageBox.Show("You must select a file")
Else
    Me.Text = F     'تغيير عنوان النافذة
    N = FreeFile()
    FileOpen(N, F, OpenMode.Input)
    Do Until EOF(N)
      S = LineInput(N)
      Buffer = Buffer & vbCrLf & S
    Loop
    txtView.Text = Buffer
    FileClose(N)
End If
```

بعد ذلك لن تحدث مشكلة إذا لم تقم باختيار ملف حيث يتم إظهار مربع رسالة لحثك على اختيار أحد الملفات.

وعلى الرغم من بساطة المثال السابق عرضه إلا أنه يوضح طريقة فعلية لاستخدام الدوارات والعبارات الشرطية و لا أظن أن الكثير من الأمثلة ستفيد هنا بل الأفضل أن تقوم بنفسك باستخدام الدوارات و العبارات الشرطية، و التعلم من أخطائك مع الرجوع لملف المساعدة كلما أمكن.